

Royal St. George’s College		Design Engineering Studio
[bookmark: _GoBack]	WIRE LEVEL EXERCISE: TC74 I2C TEMPERATURE SENSOR
Advanced Computer Engineering School		ICS3U: AVR Foundations


// PROJECT  :TC74WireAccess
// PURPOSE  :ICS3U-E: First look at Wire level access to an I2C Device 
// COURSE   :ICS3U
// AUTHOR   :C. D'Arcy
// DATE     :Created: 2023 01 27. Updated: 2025 02 20. 
// MCU      :*
// STATUS   :Working
// REFERENCE:http://darcy.rsgc.on.ca/ACES/Datasheets/TC74.pdf#page=8

#include <Wire.h>

#define TC74ADDRESS	0x4D	//0x4D(B100_1101 or 0x48(B100_1000) 
#define DURATION	1000	//pacing

void setup() {
  Serial.begin(9600);
  while(!Serial);
  Wire.begin();             		// Initiate an I2C session
  Serial.println("I2C TC74 Temperature Sensor ...");
}

void loop() {
  Wire.beginTransmission(TC74ADDRESS);// Begin transmission to the TC74
  Wire.write(0);       		// Request for data starting in Register 0
  Wire.endTransmission();             // End the transmission, then...

  Wire.requestFrom(TC74ADDRESS,1);    // Request # of bytes from stating register 
  while (!Wire.available());      	// pause & wait for the data to be sent 
  int8_t celsius = Wire.read();       // read the byte returned on the I2C bus
  float fahrenheit = celsius * 1.8 + 32;    //or round()
  
  //Publish
  Serial.print(String(celsius));      //
  Serial.write(0xC2);                 // https://www.utf8-chartable.de/ 
  Serial.write(0xB0);                 // UTF-8 degree symbol
  Serial.print("C\t");                //

  Serial.print(String(fahrenheit,1));
  Serial.write(0xC2);                 // UTF-8 degree symbol
  Serial.write(0xB0);                 //
  Serial.print("F\n");                //
  delay(DURATION);
}




